
System Administration Scripting 

Objectives After completing this lab, you will be able to: 

 Write simple scripts for managing local user accounts. 
 Write simple scripts for managing services, shared folders, and processes. 
 Write simple scripts for retrieving events from event logs. 
 Write a script for logging off the local computer. 

 

 

Prerequisites  No prerequisites are required, although knowledge of basic scripting 
fundamentals is useful. 
 
 

For more information Email scripter@microsoft.com. 

 

 

Scenario  In this lab, you will learn how scripts can be used to help manage Windows 
computers. This lab is not designed to teach you everything you need to know 
about writing scripts; that cannot be done in just one hour. However, the 
exercises will expose you to several key scripting technologies (most notably 
WMI and ADSI), as well as provide you with scripting templates that can help 
you jumpstart the process of writing your own system administration scripts. A 
number of "bonus" scripts are also included. These are scripts that are not 
officially part of the lab, but which are related to the lab exercises and might be 
of interest and use. 

The lab itself is based on the following scenario: 

One of your system administrators was recently fired. Per company policy, his 
user account was supposed to be disabled before he was told that he was being 
let go. Unfortunately, this did not happen. The administrator, who was aware 
that he might be fired, had prepared for this possibility by putting together a 
computer virus. Should he be fired, his intention was to release this virus. 
Rather than causing obvious harm, the virus was designed to provide him with 
access to all the computers on the network through several routes. That way, if 
one security breach was discovered, he could still gain access through another 
avenue. 

After being fired, the administrator was given the customary half an hour to 
clean out his office. Upon returning to his office, he discovered that he could 
still log on to the network using his administrator account. He logged on, 
released the virus, cleaned out his office, and left. 

Support personnel have discovered the virus, and identified the actions that it 
takes. This virus (which resembles no known computer virus): 

• Creates a local user named Admin2, and assigns that user full 
administrator rights to the computer. The disgruntled former 
administrator, or anyone else who is privy to the account name and 
password, can use this account to gain control over any computer on the 
network. 

• Compromises the Alerter service. If the compromised Alerter service is 
running on a computer, the former administrator can gain full access to 
that computer. If the service is running on a domain controller, he also 
gains full access to Active Directory. 

• Shares the Windows folder as PublicShare$. This gives anyone on the 
network access to the Windows folder. 

• Runs a hidden process named SystemMonitor.hta. The name is something 
of a play on words; the application does monitor the system, but not to 
check for available memory or processor use. Instead, it monitors such 
activities as the typing in of passwords, credit card numbers, and other 
confidential information. 

• Adds an entry to the Application event log stating that “New system 
monitoring software has been successfully installed.” This was probably 
added to reassure anyone who noticed that a process named 
SystemMonitor.hta had been installed on the computer. 

 

Note 

No virus has really infected the computer; the effects of this virus are all simulated. 
You will find a program named SystemMonitor.hta running on the computer, but it 
does not monitor passwords or credit card numbers. In fact, it does absolutely 
nothing.  

 

Because this is a homemade virus, your organization’s anti-virus software 
cannot help. Instead, it is up to you to analyze each of the computers under your 
control and determine whether or not they have been infected with the virus. If 
they have, you must: 

• Remove the Admin2 user account. 
• Stop the Alerter service. 
• Terminate the SystemMonitor.hta process. 
• Remove the share from the Windows folder. 
• Retrieve information from the Application event log to verify that the 

problems were the result of the virus. 
• Log off the computer. 

You will create scripts to help you carry out each of these activities. 

This lab is designed to mimic actions you might take in the event of a real 
security breach. The primary difference between this lab and real life is the fact 
that, in the lab, you will work on one standalone computer. In real life, of 
course, you would likely have to deal with hundreds or even thousands of 
computers, many of them in remote locations. This cannot be simulated in the 
lab; however, an appendix is provided that shows how the scripts you create can 
be modified to work against one or more remote computers.  

Using the Script Templates 
To help reduce the need to completely type in each script, script templates are 
provided in the C:\Scripts folder. These templates include boilerplate code that 
limits the amount of typing required to create a script. (This also helps to 
emphasize the fact that many scripts follow predictable patterns; after you learn 
how to write one WMI script, you will discover that you now know how to 
write hundreds of WMI scripts.) 

For example, suppose an exercise calls for you to write the following script: 

Set objNetwork = CreateObject("Wscript.Network") 

strComputer = objNetwork.ComputerName 

Wscript.Echo strComputer  

The template for this script might look similar to this: 

Set objNetwork = CreateObject("XXXXX.XXXXX") 

strComputer = XXXXX.XXXXX 

Wscript.Echo strComputer  

You can either type in the full script, or you can open the template, and replace 
the X's with the required information. For example, you would only have to 
type the items in bold: 
Set objNetwork = CreateObject("Wscript.Network") 

strComputer = objNetwork.ComputerName 

Wscript.Echo strComputer  

If you cannot get a script to work no matter what you try, you can find complete 
copies of all the scripts used in this lab in the C:\Solutions folder. 

Note that some of the bonus scripts will interfere with lab performance. 
 

Estimated time to 
complete this lab: 60 
minutes 

Exercise 1   
Create Scripts to Detect and Delete the Admin2, the Malicious 
User Account 

Scenario 
In this exercise, you will use an ADSI script to determine whether the Admin2 user account exists 
on the local computer. To do this, the script creates an instance of the Wscript Network object, and 
then retrieves the value of the ComputerName property (as you might expect, this is the name of the 
computer where the script is running). This value is then used to bind to the local computer and 
retrieve a list of all local user accounts. 

The script uses the WinNT provider to bind to the local computer. The provider name is case 
sensitive; it must be typed WinNT, or the script will fail. 

Then, you will use an ADSI script to remove the Admin2 user account from the local computer. To 
do this, the script connects to the local computer and then uses the Delete method to delete the 
specified user account. 

 

Tasks Detailed steps 

1. Determine whether the 
Admin2 user account exists. 

a. Double-click the Notepad shortcut on the desktop. 

b. Type the following script into the notepad.  To reduce the amount of 
typing required, you can use the template Adsi.txt. 

Set objNetwork = CreateObject("Wscript.Network") 
strComputer = objNetwork.ComputerName 
Set colItems = GetObject("WinNT://" & strComputer & 
"") 
colItems.Filter = Array("user") 
For Each objItem In colItems 
    Wscript.Echo objItem.Name  
Next 
c. Save the script as viewusers.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

"viewusers.vbs" 

d. From the command prompt, change to the C:\Scripts folder (a 
shortcut that opens a command window in C:\Scripts can be found on 
the desktop).  

e. Type cscript viewusers.vbs, and then press ENTER. 

f. Check the output to see if the account Admin2 appears in the list of 
local user accounts.  

Leave the command window open at the end of all tasks and exercises. That 
will make it faster and easier to access this window in subsequent exercises 
and tasks. 

2. Remove the Admin2 user 
account.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Adsi.txt.   

Set objNetwork = CreateObject("Wscript.Network") 
strComputer = objNetwork.ComputerName 
strUser = "Admin2" 

Set objComputer = GetObject("WinNT://" & 
strComputer & "") 
objComputer.Delete "user", strUser 
b. Save the script as deleteuser.vbs, in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:   

"deleteuser.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript deleteuser.vbs and then press ENTER. 

d. To verify that the Admin2 user account has been deleted, re-run the 
ViewUsers.vbs script created in Task 1. 

 

Why CScript? 
When you run a script in this lab, the instructions for starting that script will always be prefaced by 
the word cscript: 

cscript viewusers.vbs 

This ensures that the script runs under the CScript script host; in turn, that ensures that the output 
will appear in the command window rather than in a seemingly-endless series of message boxes (as 
would be the case using WScript, the default script host). If you do not want to type in the word 
cscript each and every time, you can change the default script host to CScript by typing the 
following at the command prompt, and then pressing ENTER: 

cscript //H:cscript 

Bonus Script: Retrieving User Accounts from Active Directory 
The preceding script, which uses the WinNT provider, retrieves user accounts from the local 
computer. What if you needed to retrieve user accounts from Active Directory? In that case, you 
need to use the LDAP provider and bind to the appropriate container in Active Directory. This 
bonus script enumerates all the user accounts found in the Users container in a hypothetical domain 
named Fabrikam.com. 

Set objOU = GetObject("LDAP://cn=users,dc=fabrikam,dc=com") 

ObjOU.Filter= Array("User") 

For each strUser in objOU.Filter 

  WScript.Echo strUser.name 

Next 

 
This script cannot be used in the lab because your computer is not a member of a domain. 

Bonus Script: Disabling a local user account 
Suppose you weren’t sure whether or not Admin2 was a valid user account. In that case, you might 
want to disable the account rather than delete it. This bonus script binds to the Admin2 user account 
on the local computer and then sets the value of the AccountDisabled attribute to True. The script 
then uses the SetInfo method to apply the change to the user account.  

SetInfo is roughly equivalent to the Save button in an application; you can change the document, 
but those changes are not permanently applied until you click Save. Likewise, you can write a 
script to change user account attributes, but those changes are not applied until you call SetInfo. 

Set objNetwork = CreateObject("Wscript.Network") 

strComputer = objNetwork.ComputerName 

Set objUser = GetObject("WinNT://" & strComputer & "/Admin2") 

objUser.AccountDisabled = TRUE 

objUser.SetInfo 

Exercise 2   
Create Scripts to Detect and Delete SystemMonitor.hta, the 
system-monitoring virus 

Scenario 
In this exercise, you will use a script to determine whether SystemMonitor.hta is running on your 
computer. To make this check, the script returns all instances of the Win32_Process class, and then 
displays the name of each process. Because HTA files run in the Mshta.exe process, you will be 
looking for any instances of Mshta.exe. 

HTA stands for HTML Application. For the most part, an HTA is simply a Web page that runs as if 
it were an executable file. You can take almost any HTM file, change the file extension to HTA, 
and then run that Web page as a standalone program. 

In this script, and in all subsequent WMI scripts, the variable strComputer (representing the name 
of the computer that the script should be run against) is set to a single dot (.). In WMI, the single 
dot, when used as a computer name, indicates that the script should run against the local computer. 
To run the script against a remote computer, simply set strComputer to the name of that remote 
computer. For example, this line of code sets the value of strComputer to FileServer01:  

strComputer = "FileServer01" 

You must have administrator rights on the remote computer in order for the script to run. 

After determining the presence of SystemMonitor.hta, you will create a script that terminates any 
running instance of SystemMonitor.hta; this is done by terminating all processes with the name 
Mshta.exe, the application responsible for running HTA programs. For each instance of Mshta.exe 
found in the collection, the Terminate method is used to kill the process.  

 

Tasks Detailed steps 

1. Determining whether 
SystemMonitor.hta is 
running.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Enumerate.txt.   

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colItems = objWMIService.ExecQuery _ 
    ("Select * from Win32_Process") 
For Each objItem in colItems 
    Wscript.Echo objItem.Name 
Next 
b. Save the script as viewprocesses.vbs in the folder C:\Scripts. To 

ensure that the .vbs file extension is used, enclose the file name in 
quotation marks before clicking Save:  

“viewprocesses.vbs” 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript viewprocesses.vbs and then press ENTER. 

d. Check the output to see if SystemMonitor.hta (Mshta.exe) is running. 

2. Terminate 
SystemMonitor.hta.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Method.txt. 

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colProcessList = objWMIService.ExecQuery _ 
    ("Select * from Win32_Process Where Name = 
'mshta.exe'") 
For Each objProcess in colProcessList 
    objProcess.Terminate 
Next 
b. Save the script as endprocess.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

"endprocess.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript endprocess.vbs and then press ENTER. 

d. To verify that SystemMonitor.hta has been terminated, rerun the 
viewprocesses.vbs script created in Task 1. 

 

Bonus Exercise: Terminating only SystemMonitor.hta 
One problem with the preceding script is that you could have legitimate HTAs that will also be 
terminated; that’s because we are terminating all instances of Mshta.exe rather than the specific 
instance in which SystemMonitor.hta is running. In Windows 2000, it is difficult, if not downright 
impossible, to identify individual HTAs; they all run under the Mshta.exe process. (This is also true 
for scripts, which all run under either Wscript.exe or Cscript.exe; however, there is no easy way to 
identify the individual script running in these processes.) In Windows XP and Windows .NET 
Server, a new property — CommandLine  — has been added to the Win32_Process class. This 
property returns the path to an individual HTA or individual script running in a generic process. 

The following bonus script includes a WQL query that specifies that only instances of 
SystemMonitor.hta should be terminated. 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colItems = objWMIService.ExecQuery _ 

    ("Select * from Win32_Process Where Name = 'mshta.exe' AND " & _ 

        "CommandLine = 'mshta.exe c:\\scripts\\systemmonitor.hta'") 

For Each objItem in colItems 

    objItem.Terminate 

Next 

Bonus Exercise: Automatically terminating SystemMonitor.hta if it 
restarts 
Some viruses are reasonably intelligent; by running a separate monitoring process or by modifying 
the registry, they find a way to restart themselves over and over again. To guard against that, you 

can create your own monitoring script, one that will check the WMI class __InstanceCreationEvent 
for any new Mshta.exe processes. The moment such a process is found, it is terminated.   

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colItems = objWMIService. _         

    ExecNotificationQuery("Select * from __instancecreationevent " _  

        & " within 1 where TargetInstance isa 'Win32_Process'") 

i = 0 

Do While i = 0 

    Set objItem = colItems.NextEvent 

    If objItem.TargetInstance.Name = "mshta.exe" Then 

        objItem.TargetInstance.Terminate 

        Wscript.Echo "An mshta.exe process was terminated at " & Now & "." 

    End If 

Loop 
 

To see how this works, start the script Monitor.vbs (found in the C:\Scripts folder): 

cscript monitor.vbs 

After the script has started, start double-click C:\Scripts\SystemMonitor.hta, and see what happens; 
the application should start, and then abruptly end. You might try starting another process (such as 
Notepad.exe or Calc.exe) and note that processes that don’t run in Mshta.exe are not affected by 
this script. 

To stop the monitoring process, either press Ctrl-C or close the command window. 

Bonus Script: Removing SystemMonitor.hta 
If SystemMonitor.hta was actually a program that breached security, you probably would not want 
to leave it installed on the computer. The following WMI script deletes the file 
C:\Scripts\SystemMonitor.hta. (Note that, in WMI syntax, you need to use double slashes to 
delineate file paths: C:\\Scripts\\SystemMonitor.hta). This script can be used not only to delete files 
on the local computer, but also to delete files on remote computers where you have administrative 
rights. To do this, simply set the value of strComputer to the name of that remote computer. 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colFiles = objWMIService.ExecQuery _ 

    ("Select * from CIM_DataFile where Name = 
'c:\\scripts\\SystemMonitor.hta'") 

For Each objFil in colFiles 

    objFile.Delete  

Next 

Exercise 3   
Create Scripts to Detect and Delete the Alerter Service that 
Allows Infiltration of Active Directory 

Scenario 
In this exercise, you will use a script to determine whether the Alerter service is running. This is 
done by connecting to the Win32_Service class and displaying the name and current state of each 
service installed on the computer. 

Next, you will use a script to stop the Alerter service. To ensure that only the Alerter service is 
stopped, the WMI query includes a filter specifying that only those services with the Name Alerter 
should be returned. After the collection of services has been returned, the StopService method is 
used to stop the service.   

 

Tasks Detailed steps 

1. Determining whether the 
Alerter service is running.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Enumerate.txt.   

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colServices = objWMIService.ExecQuery _ 
    ("Select * from Win32_Service") 
For Each objService in colServices 
    Wscript.Echo objService.DisplayName & vbTab & 
objService.State 
Next 
b. Save the script as viewservices.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

“viewservices.vbs” 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript viewservices.vbs and then press ENTER. 

d. Check the output to see if the Alerter service running. 

2. Stopping the Alerter 
service.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Method.txt. 

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colServices = objWMIService.ExecQuery _ 
    ("Select * from Win32_Service Where Name = 
'Alerter'") 
For Each objService in colServices  
    errReturnCode = objService.StopService() 
Next 
b.  Save the script as stopservice.vbs, in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

"stopservice.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript stopservice.vbs and then press ENTER. 

d. To verify that the Alerter service has stopped, wait approximately 30 
seconds (to give the service time to stop), and then rerun the 
ViewServices.vbs script created in Task 1. 

 

 

Test Yourself: Determining Free Disk Space 
Many WMI scripts, particularly those that simply retrieve information, follow a set pattern: they 
connect to the WMI service, they run a query against a specific WMI class, and then they echo the 
information returned from that query. This is the reason you can use a template to speed up your 
script writing; many of these scripts follow this same three-step approach. 

With this in mind, try writing a script that returns the amount of free disk space for all the disk 
drives on a computer (hint: use the template Enumerate.txt as a starting point). The only 
information you should need in order to create this script is this: 

• The name of the class to use is Win32_LogicalDisk. 

• The names of the two properties to use are DeviceID and FreeSpace. 
 

Bonus Exercise: Receiving notification of the success or failure of a 
script 
When you run a WMI method such as StopService(), a return code is generated that reports the 
success or failure of that operation. The scripts in this lab capture the return code, but don’t do 
anything with it. For example, this line of code runs the StopService() method, and saves the return 
code to the variable errReturnCode; however, the value assigned to errReturnCode is not referenced 
anywhere in the script: 

errReturnCode = objService.StopService() 

If you want to be notified whether or not the procedure succeeded, you can simply echo the value of 
this return code: 

Wscript.Echo errReturnCode 

In general, a value of 0 means that the procedure succeeded; any other value indicates failure. If 
you modify StopService.vbs to echo the return code and then rerun the script, you will likely see the 
value 5 returned. This indicates that the service has already been stopped. 

Bonus Script: Stopping any service on a computer  
In a scenario such as the one used in this lab, it’s useful to have a script designed solely to stop the 
Alerter service. In real life, however, it might be more useful to have a script that can stop any 
specified service. One way to do this is to create a script that allows you to pass the name of the 
service to be stopped as a command-line argument. If you named your script StopAnyService.vbs 
and you want to stop the Alerter service, you would type the following at the command prompt: 

cscript stopanyservice.vbs alerter 

If you wanted to stop the Indexing Service (which has the service name cisvc), you would start the 
script using this command: 

cscript stopanyservice.vbs cisvc 

The following script can be used to stop any service specified as a command-line argument: 

strService = Wscript.Arguments.Item(0) 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colItems = objWMIService.ExecQuery _ 

    ("Select * from Win32_Service Where Name = '" & strService & "'") 

For Each objItem in colItems  

    errReturnCode = objItem.StopService() 

Next 

Incidentally, this script has no error-handling; if you run the script without any arguments you will 
get a “Subscript out of range” error. 

But what if you run this script with more than one command-line argument (for example, cscript 
stopanyservice.vbs alerter csivc)? In that case, only the first service (alerter) will be stopped; that’s 
because the script is designed only to handle on argument (it ignores any additional arguments). To 
handle multiple arguments, you can loop through the Arguments collection, like this: 

For Each strService in Wscript.Arguments 

    Set objWMIService = GetObject("winmgmts:" _ 

        & "{impersonationLevel=impersonate}!\\" & strComputer & _ 

            "\root\cimv2") 

    Set colItems = objWMIService.ExecQuery _ 

        ("Select * from Win32_Service Where Name = '" & strService & "'") 

    For Each objItem in colItems  

        errReturnCode = objItem.StopService() 

    Next 

Next 

Exercise 4   
Create Scripts to Determine if the Windows Folder is Shared, 
then Remove that Share to Prevent Unwanted Access 

Scenario 
In this exercise, you will use a script to determine whether the Windows folder is being shared. To 
do this, the script connects to the WMI class Win32_Share and then echoes two property values for 
each share: the share name, and the local path to the folder being shared. 

Then, you will remove the share from the Windows folder. To ensure that only the share created by 
the virus is removed, a filter is included in the WMI query specifying that the only shares to be 
returned in the collection are those with the name PublicShare$. After the collection has been 
returned, that one share in the collection is removed by calling the Delete method. This method 
only removes the share; it does not delete the Windows folder.  

 

Tasks Detailed steps 

1. Determine whether the 
Windows folder is shared.    

a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Enumerate.txt.   

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colShares = objWMIService.ExecQuery("Select * 
from Win32_Share") 
For each objShare in colShares 
    Wscript.Echo objShare.Name & vbTab & 
objShare.Path    
Next 
b. Save the script as checkshares.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

"checkshares.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript checkshares.vbs and then press ENTER. 

d. Check the output to see if the Windows folder is being shared as 
PublicShare$. 

2. Insert task description here.    a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Method.txt.   

strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colItems = objWMIService.ExecQuery _ 
    ("Select * from Win32_Share Where Name = 
'PublicShare$'") 
For Each objItem in colItems 
    objItem.Delete 
Next 
b. Save the script as deleteshare.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 

marks before clicking Save:  

"deleteshare.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript deleteshare.vbs and then press ENTER. 

d. To verify that the Windows folder is no longer being shared, rerun the 
checkshares.vbs script created in Task 1. 

Exercise 5   
Create a Script to Check Event Logs 

Scenario 
In this exercise, you will retrieve specific events from the Application event log. To ensure that 
only the desired events are retrieved (as opposed to all the events in all the event logs), the WMI 
query used in the script specifies that the only events to be returned are those from the Application 
event log that have an EventCode equal to 100. 

Note: EventCode 0 is the default event code used by the LogEvent method found in 
Windows Script Host. You cannot change this event code, nor can you use a script 
to write to any event logs other than the Application log. If you need to use custom 
event codes then use event code “100” for a custom Application event. 

 
 

Tasks Detailed steps 

1. Check Event Logs.    a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Enumerate.txt.  

This script returns a number of properties for each event. For the purposes 
of this lab, however, you can shorten the script and return only the values 
for EventCode and Message (the items indicated in bold). 
strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & "{impersonationLevel=impersonate}!\\" & 
strComputer & "\root\cimv2") 
Set colItems = objWMIService.ExecQuery _ 
        ("Select * from Win32_NTLogEvent Where 
Logfile = 'Application' and " _ 
            & "EventCode = '100'") 
For Each objItem in colItems  
    Wscript.Echo "Category: " & objItem.Category 
    Wscript.Echo "Computer Name: " & 
objItem.ComputerName 
    Wscript.Echo "Event Code: " & objItem.EventCode
    Wscript.Echo "Message: " & objItem.Message 
    Wscript.Echo "Record Number: " & 
objItem.RecordNumber 
    Wscript.Echo "Source Name: " & 
objItem.SourceName 
    Wscript.Echo "Time Written: " & 
objItem.TimeWritten 
    Wscript.Echo "Event Type: " & objItem.Type 
    Wscript.Echo "User: " & objItem.User 
Next 
b. Save the script as checkevents.vbs in the folder C:\Scripts. To ensure 

that the .vbs file extension is used, enclose the file name in quotation 
marks before clicking Save:  

"checkevents.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript checkevents.vbs and then press ENTER: 

d. Check the output to see if any events with the EventCode 100 were 

found in the Application event log.  

 

Bonus Exercise: Monitoring the event logs 
The event logs often provide an early warning that a problem of some kind has either taken place or 
is about to take place. Because of that, there might be times when you would like to be notified the 
moment a particular event is recorded in one of the event logs. The following script monitors the 
event logs and issues a notice any time an event with the EventCode 100 is recorded.  

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate, (Security)}!\\" & strComputer _ 

        & "\root\cimv2") 

Set colItems = objWMIService.ExecNotificationQuery _     

    ("Select * from __instancecreationevent where TargetInstance isa " _ 

        & "'Win32_NTLogEvent' and TargetInstance.EventCode = 100 ") 

Do 

    Set objItem = colItems.NextEvent 

    Wscript.Echo "An event has been recorded in the event log." 

Loop 

To monitor the event logs, start the script MonitorLogs.vbs (found in the C:\Scripts folder): 

cscript monitorlogs.vbs  

After the script has started, run the script C:\Scripts\WriteEventToEventLog.vbs, and see what 
happens. 

To stop the monitoring process, either press Ctrl-C or simply close the command window. 

Bonus Exercise: Converting WMI dates and times  
When you ran the script CheckEvents.vbs, you might have noticed that the value of the 
TimeWritten property looked something like this: 20021008121351.660000-420. This does not 
mean that you made a mistake; WMI returns dates and times using the UTC (Universal Time 
Coordinate) format.  

Fortunately, UTC dates and times can be converted to something more readable. The following 
script retrieves the TimeWritten property for specific events, and then passes the value of that 
property to a function named WMIDateStringToDate. This function displays the UTC date-time 
value in a standard date-time format. 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colLoggedEvents = objWMIService.ExecQuery _ 

    ("Select * from Win32_NTLogEvent Where Logfile = 'Application' and " _ 

        & "EventCode = '100'") 

For Each objEvent in colLoggedEvents 

    dtmTimeWritten = objEvent.TimeWritten 

    strReturn = WMIDateStringToDate(dtmTimeWritten) 

    Wscript.Echo "Time Written: "  & strReturn 

Next 

 

Function WMIDateStringToDate(dtmTimewritten) 

    WMIDateStringToDate = CDate(Mid(dtmTimewritten, 5, 2) & "/" & _ 

         Mid(dtmTimewritten, 7, 2) & "/" & Left(dtmTimewritten, 4) _ 

         & " " & Mid (dtmTimewritten, 9, 2) & ":" & _ 

         Mid(dtmTimewritten, 11, 2) & ":" & Mid(dtmTimewritten, _ 

         13, 2)) 

End Function 

To see how this works, run the script C:\Scripts\ConvertWMIDate.vbs.  

In Windows XP and Windows Server 2003, the new SWbemDateTime object eliminates the need 
to use complicated date-time conversion functions such as this. Instead, you simply create an 
instance of the SWbemDateTime object, set the value to the TimeWritten property, and then use the 
GetVarDate method to convert the UTC date to a standard date: 

Set dtmDate = CreateObject("WbemScripting.SWbemDateTime") 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colItems = objWMIService.ExecQuery _ 

    ("Select * from Win32_NTLogEvent Where Logfile = 'Application' and " _ 

        & "EventCode = '100'") 

For Each objItem in colItems 

    dtmDate.Value = objItem.TimeWritten 

    dtmTimeWritten = dtmDate.GetVarDate 

    Wscript.Echo "Time Written: " & dtmTimeWritten 

Next 

To see how this works, run the script C:\Scripts\SWbemDateTime.vbs.  

Exercise 6   
Create a Script to log off the Computer 

Scenario 
In this exercise, you will create a script that logs you off from the local computer. This script calls 
the Win32Shutdown method, which is part of the WMI class Win32_OperatingSystem. The 
Win32ShutdownMethid is passed a value of 0, which causes the script to log off the current user. 
Other values that can be used with this method include: 

• 1 — Shutdown 

• 2 — Reboot 

• 8 — Power off 
 

Tasks Detailed steps 

1. Log Off the Computer a. Type the following script into Notepad. To reduce the amount of 
typing required, you can use the template C:\Scripts\Method.txt.  

Const LOGOFF = 0 
strComputer = "." 
Set objWMIService = GetObject("winmgmts:" _ 
    & 
"{impersonationLevel=impersonate,(Shutdown)}!\\" & 
_ 
        strComputer & "\root\cimv2") 
Set colItems = objWMIService.ExecQuery _ 
    ("Select * from Win32_OperatingSystem") 
For Each objOperatingSystem in colItems 
    objOperatingSystem.Win32Shutdown(LOGOFF) 
Next  
b. Save the script as logoff.vbs in the folder C:\Scripts. To ensure that 

the .vbs file extension is used, enclose the file name in quotation marks 
before clicking Save:  

"logoff.vbs" 

c. From the command prompt, change to the C:\Scripts folder (if 
necessary). Type cscript logoff.vbs and then press ENTER.  

Note that when you run this script, you will be logged off.  

 

 

For More Information 
If you would like to learn more about system administration scripting, point your Web browser 
towards http://www.microsoft.com/technet/scriptcenter. The TechNet Script Center features 
hundreds of scripts devoted to system administration; in addition, the Script Center provides links 
to numerous scripting resources. 

Appendix A: Running a Script Against All the Computers Listed 
in a Text File 
One way to run a script against multiple computers is to read a list of computers names from a text 
file. In the script shown below, the FileSystemObject is used to read a list of server names from a 
text file; the script then runs against each of these computers, reporting the number of services 
installed on that computer.  

The text file itself should simply be a list of computer names, similar to this: 

atl-dc-01 

atl-dc-02 

atl-dc-03 

 

Note: This might not be the most practical use of an enterprise script; you rarely need to know how 
many services are installed on a computer. However, the actual activity performed by the script is 
irrelevant; the important aspect is how the script reads server names from a text file and then 
connects to each computer. A trivial task such as reporting the number of services installed was 
chosen simply to keep the focus on connecting to multiple computers and not on the activity carried 
out after each connection is made. 

Const ForReading = 1  

Set objFSO = CreateObject("Scripting.FileSystemObject") 

Set objTextFile = objFSO.OpenTextFile _ 

    ("c:\scripts\servers.txt", ForReading) 

Do Until objTextFile.AtEndOfStream  

    StrComputer = objTextFile.Readline 

    Set objWMIService = GetObject("winmgmts:" _ 

        & "{impersonationLevel=impersonate}!\\" & strComputer & _ 

            "\root\cimv2") 

    Set colServices = objWMIService.ExecQuery _ 

        ("Select * from Win32_Service") 

    Wscript.Echo strComputer, colServices.Count 

Loop 

 

Appendix B: Running a Script Against All the Computers in an 
Active Directory Container 
You can also configure a script to run against every computer in a domain or Active Directory 
container. The script shown below returns a list of computer names from the Domain Controllers 
OU in a hypothetical domain named Farbrikam.com; the script then runs against each of these 
computers, again reporting the number of services installed on that computer.  

Set objOU = GetObject("LDAP://OU=Domain Controllers, DC=fabrikam, DC=com") 

objOU.Filter = Array("Computer") 

For Each objComputer in objOU  

    strComputer = objComputer.CN 

    Set objWMIService = GetObject("winmgmts:" _ 

        & "{impersonationLevel=impersonate}!\\" & strComputer& "\root\cimv2") 

    Set colServices = objWMIService.ExecQuery _ 

        ("Select * from Win32_Service") 

    Wscript.Echo strComputer, colServices.Count 

Next 

 

Appendix C: Scripts Used to Set Up This Lab 
The following scripts were all used to help set up the conditions under which this lab was run; for 
example, there are scripts here that share the Windows folder, that create a user account named 
Admin2, and that start SystemMonitor.hta in a hidden window.  

Create a local user account named Admin2 
Set objNetwork = CreateObject("Wscript.Network") 

strComputer = objNetwork.ComputerName 

Set colAccounts = GetObject("WinNT://" & strComputer & "") 

Set objUser = colAccounts.Create("user", "Admin2") 

objUser.SetInfo 

 

Share the Windows folder 
strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colOperatingSystems = objWMIService.ExecQuery _ 

    ("Select * from Win32_OperatingSystem") 

For Each objOperatingSystem in colOperatingSystems 

    strWindowsFolder = objOperatingSystem.WindowsDirectory 

Next 

Const FILE_SHARE = 0 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set objNewShare = objWMIService.Get("Win32_Share") 

errReturn = objNewShare.Create _ 

    (strWindowsFolder, "PublicShare$", FILE_SHARE, ,_ 

        "Public share for the Finance group.") 

 

Write an event to the event log 
Const EVENT_SUCCESS = 0 

Set objShell = Wscript.CreateObject("Wscript.Shell") 

objShell.LogEvent EVENT_SUCCESS, _ 

    "System monitoring software successfully installed." 

 

Start SystemMonitor.hta in a hidden window 
Const HIDDEN_WINDOW = 12 

strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set objStartup = objWMIService.Get("Win32_ProcessStartup") 

Set objConfig = objStartup.SpawnInstance_ 

objConfig.ShowWindow = HIDDEN_WINDOW 

Set objProcess = GetObject("winmgmts:root\cimv2:Win32_Process") 

errReturn = objProcess.Create _ 

    ("mshta.exe c:\scripts\SystemMonitor.hta", null, objConfig, intProcessID) 

 

Start the Alerter service 
strComputer = "." 

Set objWMIService = GetObject("winmgmts:" _ 

    & "{impersonationLevel=impersonate}!\\" & strComputer & "\root\cimv2") 

Set colServices = objWMIService.ExecQuery _ 

    ("Select * from Win32_Service Where Name = 'Alerter'") 

For Each objService in colServices  

    errReturnCode = objService.StartService() 

Next 

Create shortcuts on the desktop 
Set objShell = WScript.CreateObject("WScript.Shell") 

strDesktopFolder = objShell.SpecialFolders("AllUsersDesktop") 

Set objShortCut = objShell.CreateShortcut(strDesktopFolder & _ 

     "\Notepad.lnk") 

objShortCut.TargetPath = "Notepad.exe" 

objShortCut.Description = "Starts Notepad." 

objShortCut.WorkingDirectory = "C:\Scripts" 

objShortCut.Save 

Set objShortCut = objShell.CreateShortcut(strDesktopFolder & _ 

    "\Command Prompt.lnk") 

objShortCut.TargetPath = "Cmd.exe" 

objShortCut.Description = "Opens the command prompt." 

objShortCut.WorkingDirectory = "C:\Scripts" 

objShortCut.Save 

 

Create the folder C:\Scripts 
Set objFSO = CreateObject("Scripting.FileSystemObject") 

Set objFolder = objFSO.CreateFolder("C:\Scripts") 

 

Add an entry to the registry 
Set objShell = CreateObject("Wscript.Shell") 

strValue = "wscript ""C:\Setup\createprocess.vbs""" 

strEntry = "HKLM\Software\Microsoft\Windows\CurrentVersion\Run\MEC" 

errReturnCode = objShell.RegWrite(strEntry, strValue) 

 

Copy the lab files 
Const OverwriteExisting = True 

Set objFSO = CreateObject("Scripting.FileSystemObject") 

objFSO.CopyFile "*.txt" , "C:\Scripts\" , OverwriteExisting 

objFSO.CopyFile "*.hta" , "C:\Scripts\" , OverwriteExisting 

objFSO.CopyFile "Monitor.vbs" , "C:\Scripts\" , OverwriteExisting 

objFSO.CopyFile "MonitorLogs.vbs" , "C:\Scripts\" , OverwriteExisting 

objFSO.CopyFile "ConvertWMIDate.vbs" , "C:\Scripts\" , OverwriteExisting 

objFSO.CopyFile "*ViewUsers.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*DeleteUser.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*ViewProcesses.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*EndProcess.vbs" ,  "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*Monitor.vbs" ,  "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*ViewServices.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*StopService.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*CheckShares.vbs" ,  "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*DeleteShare.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*CheckEvents.vbs" , "C:\Solutions\" , OverwriteExisting 

objFSO.CopyFile "*MonitorLogs.vbs" , "C:\Solutions\" , OverwriteExisting 

   

Delete a folder 
Set objFSO = CreateObject("Scripting.FileSystemObject") 

If objFSO.FolderExists("C:\Scripts") Then 

    objFSO.DeleteFolder("C:\Scripts") 

End If 

 

Hide a folder 
Set objFSO = CreateObject("Scripting.FileSystemObject") 

Set objFolder = objFSO.GetFolder("C:\Setup") 

If objFolder.Attributes = objFolder.Attributes AND 2 Then 

    objFolder.Attributes = objFolder.Attributes XOR 2  

End If  

 

 

 


